﻿CREATE PROCEDURE [dbo].[ShiftInDocInsert]
(--приход по внутреннему перемещению
	@Id int output,
	@DocDate smalldatetime,
	@DocNumber varchar(20) output,
	@ShiftDepartmentId int,  --код подразделения, откуда пришел товар
	@ContragentDocDate smalldatetime, --дата документа исходного подразделения
	@ContragentDocNumber varchar(20), --номер документа исходного подразделения
	@ParentId int,										--идентификатор документа исходного подразделения
	@Note varchar(250),
	@MD datetime output,
	@MW varchar(50)
)
AS
DECLARE @strMsg nvarchar(500), @intRet int, @intErrCode int, @intNextId int
DECLARE @DocTypeId int,	@DepartmentUserId int
	SET NOCOUNT ON

IF (@ShiftDepartmentId Is Null) OR (@ParentId Is Null)
BEGIN
	SELECT @intErrCode = 1, @strMsg = 'Неправильные параметры в вызове процедуры ' + Object_Name(@@Procid) + ' !'
	RAISERROR (@strMsg, 11, @intErrCode)
	RETURN @intErrCode
END

IF @DocNumber Is Null
BEGIN
	EXEC [dbo].[GetNextDocNumber] 5, @intRet OUT
	SET @DocNumber = cast( @intRet as varchar(20))
END

IF @DocDate Is Null SET @DocDate = dbo.TodayDate() 
ELSE SET @DocDate = dbo.NormalizeDate(@DocDate)

IF (@ContragentDocDate Is NOT Null) SET @ContragentDocDate = dbo.NormalizeDate(@ContragentDocDate)

IF (@ContragentDocDate Is NOT Null) AND @ContragentDocDate > @DocDate
BEGIN
	SELECT @intErrCode = 2, 
	@strMsg = 'Дата документа исходного подразделения больше даты приходной накладной!'
	RAISERROR (@strMsg, 11, 1)
	RETURN @intErrCode
END

SELECT @DepartmentUserId = Id FROM CD.DepartmentUsers WHERE Login = suser_sname()

EXEC [dbo].[ProcessStatistic] @Note OUTPUT, @MD OUTPUT, @MW OUTPUT
--получение значения идентификатора
EXEC @intRet = dbo.GetNextCounterFor 'Documents', @intNextId OUTPUT
IF @intRet <> 0 RETURN @intRet

	SET NOCOUNT OFF;
INSERT INTO [CA].[Documents] (Id, [DocTypeId], [DocDate], [DocNumber], 
ShiftDepartmentId, ContragentDocDate, ContragentDocNumber, DepartmentUserId, ParentId,
[Note], [MD], [MW]) 
VALUES (@intNextId, 5, @DocDate, @DocNumber, 
@ShiftDepartmentId, @ContragentDocDate, @ContragentDocNumber, @DepartmentUserId, @ParentId,
@Note, @MD, @MW);
	
select @Id = @intNextId
--SELECT Id, GId, DocTypeId, DocDate, DocNumber, 
--DepartmentId, ContragentDocDate, ContragentDocNumber, 
--DepartmentUserId, Note, MD, MW 
--FROM CA.Documents WHERE (Id = SCOPE_IDENTITY())

RETURN 0

